

FLOATING IN-OUT ROUTINES 2/24/67 (FINOUT,14)

/APR 8-64, MOD SRW 12-65

/FLIP 10T

```

FLIP,      LAC I (FSA)
FLIP+1,    DAC A          /KEEP TEXT PTR IN A
           D2M G          /HIGH ORDER MANTISSA IN G
           D2M H          /LOW ORDER MANTISSA IN H
           D2M C          /CHARACTERISTIC IN C
           D2M I (ATEM)   /CORRECTIVE SCALE FACTOR IN ATEM
           D2M B          /EXPONENT OF "E" IN B

CHLPLS,
FLIP1,     LCH I A        /MAIN LOOP, CLASSIFY CHARACTER
           SAD (CHARAC L,)
           JNP FLIP4
           SAS CHLPLS      /((CHARAC L+))
           SAD (CHARAC L-)
           JNP FLIP5
           SEA I
           JNP FLIP5
           SAD CHARLE      /((CHARAC LE))
           JNP FLIP6
           RAL 6S          /TEST FOR A DIGIT
           XOR B13         /((2))
           SUB CLP,
           SPA             /WAS CHARACTER A DIGIT?
           JNP FLIP7       /YES, GO PROCESS IT,
           SZF 4           /ANY DIGITS IN THIS FIELD?
           JNP FLIP13      /YES, GO EVALUATE FOREGOING STRING,
FLIP2,     JSP FLIPX       /EXIT SUBROUTINE, FAILURE
FLIP3,     JNP R1         /GIVE R1 TO INDICATE FAILURE

FLIP4,     SZF I 1        /DECIMAL POINT, CHECK FOR REDUNDANCE
           SZF 2          /HAVE WE SEEN AN "E" ALREADY?
           JNP FLIP2      /YES, THIS CHARACTER FAILS,
           STF 1          /FLAG 1 INDICATES "SEEN A DECIMAL POINT"
           JNP FLIP1      /GO GET NEXT CHARACTER

FLIP5,     SZF I 3        /SIGN, CHECK FOR REDUNDANCE
           SZF 1          /HAVE WE SEEN A DECIMAL POINT ALREADY?
           JNP FLIP2      /YES, THIS CHARACTER FAILS,
           SZF 4          /HAVE WE ALREADY SEEN A DIGIT?
           JNP FLIP13     /YES, THIS CHARACTER FAILS
           STF 3          /INDICATES "SEEN A SIGN" IN THIS FIELD
           SAS (CHARAC L-)
           JNP FLIP1      /NO, GET NEXT CHARACTER,
           SZF 2          /SET FLAG 6 IF WE HAVE SEEN AN E
           STF 6          /FLAG 6 INDICATES A NEGATIVE EXPONENT OF "E"
           SZF I 2        /IF FLAG 2 IS NOT SET, SET FLAG 5
           STF 5          /FLAG 5 INDICATES A NEGATIVE MANTISSA
           JNP FLIP1      /GET NEXT CHARACTER

```

```

FLIP6,  SZF 1 2      /FOUND AN E, CHECK FOR PEDUNDANCY
        SZF 1 4      /HAS THE E PRECEDED ALL DIGITS?
        JMP FLIP2     /YES, THIS CHARACTER FAILS
        CLF 1        /INITIALIZE FOR A NEW FIELD
        STF 2        /FLAG 2 INDICATES "SEEN AN "E""
        CLF 3        /DONE SO A SIGN WON'T FAIL
        CLF 4
        JMP FLIP1     /GET NEXT CHARACTER

FLIP7,  STF 4        /FOUND A DIGIT
        ADD C10,     /GETS VALUE OF DIGIT IN AC
        DAC E        /SAVE DIGIT IN E TEMPORARILY
        SZF 2        /HAVE WE SEEN AN "E"
        JMP FLIP12    /YES, GO BUILD UP EXPONENT OF "E",
        JSP FPM10     /NO, BUILD UP MANTISSA
        JSP FPNORM    /PREPARE TO ADD IN THIS DIGIT
        DIO H
        LAW 4        /NEEDED TO FINISH MUL BY 10
        ADD C
        SUB F        /SUBTRACT DECREMENT LEFT BY NORMALIZE ROUTINE
        DAC C        /SAVE NEW CHARACTERISTIC
        LAC E        /MAKE THIS DIGIT INTO A MANTISSA
        RAP 55
        SCI,USZL 1   /TEST FOR 0 & CLEAR 10
        JMP FLIP8     /PREVIOUS MANTISSA WAS NOT ZERO, GO AROUND,
        DAC G        /STORE CURRENT DIGIT AS MANTISSA
        LAW 4        /WITH 4 AS CHARACTERISTIC
        DAC C
        JMP FLIP11    /GO CHECK FOR DIGIT TO RIGHT OF DEC. PT.

FLIP8,  DAC E        /SAVE HI-ORD PART OF MANTISSA, LO-ORD=0
        LAW 4        /CHAR.=4, SET UP TO UN-NORMALIZE FOR ADDING
        DAC F

FLIP9,  SAD C        /UN-NORMALIZING LOOP, FINISHED WHEN C=F,
        JMP FLIP10    /SHIFT MANTISSA RIGHT 1 BIT
        LAC E
        SCP 15
        DAC E
        IDX F        /ADD 1 TO CHARACTERISTIC TO PRESERVE VALUE
        JMP FLIP9     /GO AROUND AGAIN

FLIP10, LAC E        /ADD THE TWO 36-BIT MANTISSAS TOGETHER
        SXP
        TAD H
        SWP
        TAD G
        SCP 15
        AND (377777) /NUMBERS ARE +, CORRECT POSSIBLE OVERFLOW
        DAC G        /SET UP NORMALIZE ROUTINE
        JSP FPNORM    /GO NORMALIZE
        DIO H        /SAVE LO-ORD MANTISSA, (MUST BE NON=0)
        IDX C        /OFFSET UN-NORMALIZING SHIFT
        SUB F        /SUBTRACT NORMALIZING DECREMENT
        DAC C        /STORE NEW CHARACTERISTIC

FLIP11, SZF 1        /ARE WE TO THE RIGHT OF THE DECIMAL POINT?
        IDX I (ATEM) /YES, IDX THE CORRECTIVE SCALE FACTOR
        JMP FLIP1     /GET NEXT CHARACTER

```

```

FLIP12,  LAC B           /GET CURRENT EXPONENT OF E
          MUL C10,        /GET LOW-ORDER PART OF PRODUCT
          SWP             /OVERFLOWS IF LARGER THAN 17 BITS
          SNI+USMA I      /OVERFLOW
          JMP FLIP1B
          SAR 1S          /MULTIPLY LEAVES THINGS SHIFTED LEFT A BIT
          ADD E           /ADD IN CURRENT DIGIT
          DAC B           /SAVE NEW EXPONENT OF "E"
          JMP FLIP1       /GET NEXT CHARACTER

FLIP13,  IDX I (USERPC)  /SUCCESS, SET UP R2
          LAC B           /GET EXPONENT OF "E"
          SZA+USZF 6      /RESTORE SIGN UNLESS 0
          CMA
          SUB I (ATEM)    /COMBINE IT WITH CORRECTIVE SCALE FACTOR
          DAC I (ATEM)    /POWER OF TEN TO TAKE NUMBER TO
FLIP14,  LAC I (ATEM)    /GET CORRECTIVE SCALING FACTOR
          SZA I           /END TEST, QUIT WHEN CORRECTION = 0
          JMP FLIP17      /EXIT LOOP, NUMBER IS NOW OK
          SPA            /MUST WE MUL OR DIV TO GET CORRECTION TO 0?
          JMP FLIP16      /GO ADD 1 TO SCALE FACTOR AND DIVIDE BY TEN
          SUB B17         /SUB (1) AND DIV BY 10
          DAC I (ATEM)
          JSP FPM10       /FPM10 MULTIPLIES MANTISSA IN G AND H BY TEN
          JSP FPNORM      /GO NORMALIZE
          DIO H           /SAVE LOW-ORDER MANTISSA IN H
          LAR 4           /ADD (4) TO CHARACTERISTIC IS PART OF MULTIPLY
FLIP15,  ADD C           /ADD IN PREVIOUS CHARACTERISTIC
          SUB F           /SUBTRACT DECREMENT LEFT BY NORMALIZE ROUTINE
          DAC C           /SAVE NEW CHARACTERISTIC
          S&L I           /TEST FOR A ZERO MANTISSA
          JMP FLIP14      /MANTISSA WAS NON-ZERO, GO AROUND AGAIN,
          JSP FLIPX       /MANTISSA WAS ZERO, PREPARE TO EXIT
          JMP FADD6       /GO PUT ZERO IN FLOATING AC AND RETURN TO USER

FLIP16,  IDX I (ATEM)    /ADD 1 TO SCALE FACTOR & DIV BY 10
          JSP FPD10       /DIVIDE MANTISSA IN G AND H BY TEN
          JSP FPNORM      /GO NORMALIZE
          DIO H           /SAVE LOW-ORDER MANTISSA
          LAR I 3         /DECREMENT CHARACTERISTIC FINISHES DIV BY 10
          JMP FLIP15      /COMPUTE NEW CHARACTERISTIC & LOOP AROUND
FLIP17,  LAC G           /NUMBER IS NOW OF PROPER MAGNITUDE
          LIO H           /GET MANTISSA
          S&F 5           /IF FLAG 5 IS SET, MANTISSA IS NEGATIVE
          CMA+UCM1
          DAC G           /SAVE MANTISSA WITH CORRECT SIGN
          DIO H
          JSP FLIPX       /PREPARE TO EXIT
          LIO H           /GET LOW-ORDER MANTISSA IN I.O.
          JMP FADD4       /GO PACK # INTO FAC & RETURN
FLIP18,  IDX I (USERPC)  /OVERFLOW
          JSP FLIPX       /PREPARE TO EXIT
          JMP FPOVFL      /GO TO THE FLOATING OVERFLOW ROUTINE

```

```

FLIPX,   DAP D           /FLIP EXIT SUBROUTINE
        LAC CJ
        DIP D           /SAVE RETURN IN D
        LAW I 1         /GET PTR TO CHAR AND UNSTEP IT
        ADD A
        IOC
        IOC
        DAC I (FSA)     /STORE PTR IN FSA
        JMP D           /RETURN

```

/FLOPE & FLOPF IOTS

```

FLOP,    LIO I (JMODE)  /CHECK FOR JBH
        SPI+USZF 6
        JMP R2
        LAC I (FAC+1)   /GET FL PNT #
        LIO I (FAC)
        SPI             /MAKE SURE MANTISSA IS POSITIVE
        CHA+UCM1+USTF 1 /NEGATIVE FLAG
        AND (777420)    /GET RID OF CHARACTERISTIC
        DIO G           /SAVE 36-BIT POSITIVE MANTISSA IN G AND H
        DAC H
        LAW I 1         /C CONTAINS EXPONENT OF 10 LESS 1
        DAC C
        LAC I (FAC+1)   /GET CHARACTERISTIC
        RAP BS
        SAR 90
        SAR 10
FLOP1,   DAC E           /SAVE IT IN E
        SAN             /IS CHARACTERISTIC >0?
        JMP FLOP6       /YES, DIVIDE FL, PNT, # BY 10 & IDX C
        ADD (3)
        SGA             /TEST WHETHER NUMBER IS <=3
        JMP FLOP7       /EXIT, CHARACTERISTIC IS =3, =2, =1, OR 0.
FLOP2,   JSP FPM10      /MUL FL, PNT, # BY 10 & DECREMENT C
        JSP FPNORM      /NORMALIZE
        DIO H           /LO-ORD MANTISSA
        SZL I           /ZERO?
        JMP FLOP4       /NO, NON-ZERO MANTISSA. GO AROUND.
FLOP3,   DZM C          /GO PRINT A 0
        JMP FLOP10      /EXIT WITH C, G, AND H ALL PLUS ZERO
FLOP4,   LAW I 1        /SUBTRACT 1 FROM EXPONENT OF TEN
        ADD C
        DAC C
        LAW 4
FLOP5,   ADD E          /PART OF MUL BY 10
        SUB F           /ADD IN CHARACTERISTIC
        JSP F           /SUBTRACT DECREMENT LEFT BY NORMALIZE ROUTINE
        JMP FLOP1       /GO AROUND AGAIN

```

```

FLOP6,   JSP FPO10      /DIVIDE NUMBER BY TEN
        JSP FPNORM      /NORMALIZE
        DIO H          /SAVE LOW ORDER HALF OF MANTISSA
        SZL            /WERE WE NORMALIZING PLUS ZERO?
        JMP FLOP3      /YES, EXIT, PRINT OUT A ZERO,
        IDY C          /ADD 1 TO EXPONENT OF TEN
        LAM I 3        /PART OF DIV BY 10
        JMP FLOP5      /COMPUTE NEW CHARACTERISTIC & GO AROUND AGAIN

FLOP7,   SZF 3          /HAVE WE BEEN HERE BEFORE?
        JMP FLOP8      /YES, GO ON TO NEXT PART OF ROUTINE,
        STF 3          /REMEMBER THAT WE'VE BEEN HERE
        JMP FLOP2      /SOME -3 CHARACTERISTICS CAN GO TO 0, TRY

FLOP8,   LIO H          /FP # IN E,C,H; UN-NORMALIZE G,H TO GET E=0
        LAC E
FLOP9,   SZA I          /FINISHED?
        JMP FLOP10     /EXIT
        LAC G
        SCR 15
        DAC G
        IDX E
        JMP FLOP9

FLOP10,  DIO H          /PUT LOW ORDER MANTISSA BACK IN H
        LAM ATEM
        DAC A          /A IS TEN STORAGE PTR
FLOP11,  JSP FPM10      /LOOP TO GENERATE 8 DIGITS
        LAC G          /HI-ORD RESULT OF MUL
        DAC B
        AND (17777)    /REMOVE BITS TO LEFT OF BIN PT
        SCL 4S         /PUT BIN PT BETWEEN BITS 0&1
        DAC G          /STORE NEW MANTISSA
        DIO H
        LAC B          /GET NEW DIGIT ALONE IN LO END OF AC
        RAL 5S
        AND (17)
        DAC I A        /STORE IN TEN AREA
        IDY A          /MAKE A POINT TO NEXT REGISTER
        SAS (ATEM+10)  /QUIT AFTER 8 DIGITS
        JMP FLOP11     /GO AROUND AGAIN
        LAC I (STS)
        DAC G          /SET UP BYTE PTR FOR TEXT
        IDY I (USERPC) /GET PTR TO FORMAT WORD
        DZM E
        DAP E          /SAVE PTR
        LCH I E        /1ST BYTE OF FORMAT WD IS # OF COLUMNS
        RAL 6S
        IDA
        CMA
        DAC I (ATEM+10) /COLUMN COUNT WD
        LCH I E        /NEXT BYTE IS # OF SIGNIFICANT FIGURES
        RAL 6S
        DAC B
        LCH I E        /3RD BYTE IS # OF PLACES PAST DEC. PT.
        RAL 6S
        IDA
        DAC F          /1 * # OF SIG. FIGURES

```

FORM 6510

PRINTED IN U.S.A.

```

LAC C          /THIS SECTION CALCULATES WHICH DIGIT TO ROUND
SZF 2          /SKIP ON F FORMAT
CLA
ADD F
LIA
SUB B
SZM           /ROUND ON ACCURACY OR DIGITS?
LID B         /ACCURACY
LAT           /# OF CHARACTERS TO BE TYPED
SPA          /TEST FOR ROUND BEING TOO FAR LEFT
STF 4         /TOO FAR LEFT, PF4 PREVENTS ROUNDING
SUB B14       /(8,)
SMA          /TEST FOR ROUND BEING TOO FAR RIGHT
STF 4         /PREVENT ROUNDING IF SO
ADD (ATEM+10) /GETS PTR AT REG TO ROUND
DAC A         /SAVE PTR FOR LOOP AT FLOP12
DAC E         /SAVE PTR MORE PERMANENTLY FOR USE LATER
LAW I 5       /SUB (5) & TEST WHETHER +0=
ADD I 4
SZF I 4       /SKIP TO SUPPRESS ROUNDING
SPA          /SKIP TO ADD CARRY TO NEXT CHARACTER
JMP FLOP14    /QUIT ROUNDING

CHRLEQ,
FLOP12,      DCM I A          /CARRY LOOP ON ROUND=UP
LAW I 1      /UNSTEP TEM PTR
ADD A
CHARL4,      DAC A
SAS (ATEM-1) /TEST WHETHER ROUND CARRIED OFF THE LEFT END
JMP FLOP13   /OTHERWISE, GO AROUND
IDX I (ATEM) /ATEM WAS 0, MAKE IT 1
IDX C        /ADD 1 TO EXPONENT OF TEN
IDX E
JMP FLOP14   /EXIT FROM LOOP

CHARLE,
FLOP13,      IDX I A          /ADD CARRY FROM PREVIOUS DIGIT
SAP C10,     /DOES THIS CHARACTER CARRY LEFT NOW?
JMP FLOP12   /GO AROUND CARRY LOOP AGAIN
FLOP14,      CLI             /ROUNDOFF NOW COMPLETE, READY TO PRINT
SZF 1        /FLAG 1 SET FOR NEGATIVE NUMBER
LID (CHARAC L-)
IDX I (ATEM+10)
SZF I 5
JSP TYOIF
LAW I 2      /INITIALIZE D FOR STORING DIGITS
SZF I 2
SUP C
SMA
LAW I 1
DAC D        /D GETS IDX'ED FOR EACH DIGIT, STORE "," AT 0
LAC C        /INITIALIZE B
SZF I 2
SMA
LAW I 1
ADD (ATEM)
DAC B        /B ALWAYS POINTS TO NEXT DIGIT, AFTER IDX

```

```

FLOP15,  IDX D          /LOOP FOR PRINTING
        SZA           /SKIP IF TIME TO PRINT DECIMAL POINT
        JMP FLOP16    /OTHERWISE, GO AROUND
        LID (CHARAC L) /PRINT DEC PT
        JMP FLOP20

FLOP16,  SAA F          /ALL DONE PRINTING DIGITS?
        JMP FLOP21    /QUIT PRINTING DIGITS, EXIT FROM LOOP
        IDX B          /GET PTR TO NEXT DIGIT TO BE PRINTED
        SUB (ATEM)     /FIGURE OUT IF TOO FAR LEFT
        SNA           /TOO FAR LEFT?
        JMP FLOP18    /NO, GO AROUND,
FLOP17,  LAW 20         /YES, PRINT A 0
        JMP FLOP19

FLOP18,  LAC B          /FIGURE OUT IF TOO FAR RIGHT
        SUB E          /E IS PTR TO DIGIT BOUNDED AT
        SNA           /TOO FAR RIGHT?
        JMP FLOP17    /YES, PAST SIGNIFICANT FIGURES, PRINT A 0
        LAW 20         /NO, GET INTERNAL CODE FOR THIS DIGIT,
        ADD I B

FLOP19,  RGR 6S         /POSITION INTERNAL CODE FOR PRINTING
FLOP20,  IDX I (ATEM+10)
        SEF I 5
        JSP TYOIF
        JMP FLOP15    /GO AROUND LOOP AGAIN
FLOP21,  SEF I 2         /FINISHED WITH DIGITS, TEST FOR E FORMAT
        JMP FLOP22    /"E" FORMAT, GO AROUND,
        LAW 4
        ADD I (ATEM+10)
        DAC I (ATEM+10)
        SEF 5
        JMP FLOP22
        LID CHARLE     /((CHARAC LE) E FORMAT, PRINT THE E
        JSP TYOIF
        LAC C          /GET EXPONENT OF E, TEST SIGN, AND TYPE IT
        LID (CHARAC L)
        SNA
        CLI+UCMA
        DAC A
        JSP TYOIF
        LAC A
        CLI+UCMA+USWP
        RCL 1S
        DIV C10,
05,      5             /CONSTANT STORED AFTER DIVIDE
        RJP 6S
        RCL 6S
        IOP C2020
        RUP 6S
        RGR 6S         /POSITION FIRST DIGIT
        JSP TYOIF
        JSP TYOIF

```



```

FLOP22,  LAC G          /UPDATE PTR
        SZF I 6
        DAC I (STS)
        SZF 5
        JMP FLOP23      /RIGHT ADJUST
        LAC ML4
        SZF I 6
        DCH I 6
        JMP R1

FLOP24,  CLI
        JSP TYCIF
FLOP23,  ISP I (ATEM+10)
        JMP FLOP24
        CLF 5
        JMP FLOP14

/SUBR TO MUL A 36BIT +MANTISSA IN G,H BY 10 & SHIFT BIN PT 4 TO RIGHT
FPM10,  DAP D
        LAC CJ          /RETURN
        DIP D
        LAC H          /GET MANTISSA, WITH HALVES REVERSED
        LIO G
        AND (777760)    /REMOVE LAST 4 BITS TO KEEP SIGN +
        RCR 33          /UN=NORM,INC SHIFT TO PROTECT SIGN
        DIO G
        RAP 15
        MUL C5          /(5), MUL (5) & SHIFT 1
        DAC F
        DIO H
        LAR 5
        MUL G
        RCR 15
        LAT
        ADD F          /COMBINE THE TWO HALVES OF THE NEW MANTISSA
        DAC G
        LIO H          /LEAVES LO-ORD IN H & IO, HI-ORD IN G
        JMP D          /RETURN

/SUBR TO DIV A 36BIT +MANTISSA IN G,H BY 10 & SHIFT BIN PT 3 TO LEFT
FPD10,  DAP D
        LAC CJ
        DIP D
        LAC G          /SET UP FOR DIVIDE
        LIO H
        SCR 15          /SIC,
        DIV CHARL4      /(2400000)
C1849,, 1849,
        DAC G
        CLA+USWP
        DIV CHARL4      /(2400000)
C720,, 720,
        PAL 15
        DAC H
        LIA
        JMP D          /SAME EXIT FORMAT AS FPM10
                        /RETURN

```

START

