```
DISPATCHER (DISPAT,104) 10/2/73  JGC/BPC/JBL

/CHAN 16 BREAK
DISPAT,   LIF"U"SCF"U"CLL
          DIO DSP1        /SAVE FLAGS
          RCL 2S          /TOP OF IO CLEAR
          EEM
          SAD ([DDTS6+3]"T"4)        /IS IDDT "EXECUTING" INSTR?
          JMP DSP3        /YES, GO ADJUST PC FOR CALLING SEQUENCES
DSP4,     RTB
          DIO TRPBUF
          CLA"U"SWP
          SMA             /AN IOT?
          JMP ELGIOT      /NO
          RAL 5S          /YES, TEST I BIT
          SPA
          JMP DSP16       /YES, CORE 16 IOT
          RAL 1S
          DCH (JMP 140000 DSP2)      /REGULAR DISPATCH
          RCL 6S
          LFI
DSP2,     JMP I DSPTB+0

DSP3,     LAW I 1         /DDT PC ADJUST
          ADD I (DDTS6+4)
          DAP 71
          JMP DSP4

DSP16,    RAL 1S          /CORE 16 DISPATCH
          DCH (JMP 140000 DSP5)
          RCL 6S          /GET REDISPATCH BITS INTO IO
          LFI
          LAC 70          /MOVE AC,PC,IO
          DAC I (C16AC)
          LAC 71
          DAC I (C16PC)
          LAC 72
          DAC I (C16IO)
          LAC DSP1        /FLAGS, TOO
          DAC I (C16FLA)
          LAC I (BITS)
          AND (-40000)    /RESTART MODE DEBREAK LEGAL BIT
          DAC I (BITS)
DSP5,     LAC C16TB+0     /GET DEBREAK DISPATCH
DSP12,    DAC 71          /DEBREAK, DISPATCH
          JMP I 71

DSP1,     0               /FLAGS
TRPBUF,   0
```

```
ILGIOT,   EEM
ELGIOT,   LIO I (BITS)    /ILLEGAL INSTR.
          LAI
          AND (-500000)   /IDDT AND RESTART BITS
          DAC I (BITS)
          SPI I           /IS GUY UNDER IDDT
          JMP DSP8        /NO
          LAC C16BKP      /BREAKPOINT, GET DDT
DSP9,     LIO 71          /SET UP CRASH AC,PC,IO,C16 TEMP STORAGE
          DAC 71
          DIO I (HH16PC)
          JSP C16TSV
          LIO 70
          DIO I (HH16AC)
          LIO 72
          DIO I (HH16IO)
          LIO DSP1
          DIO I (HH16FG)
          LIO TRPBUF
          JMP I 71

DSP8,     RIL 1S
          SPI
          JMP XDDTBK      /HELD BY XDDT + CRASHED
          LAC TRPBUF
          SAS (100400)    /HLT?
          JMP DSP10
          RIL 3S
          SPI
          JMP REALHALT
          LAC (20000)     /REALHALT LEGAL BIT
          IOR I (BITS)
          DAC I (BITS)
          LAC C16HLT
          JMP DSP12

DSP10,    LAC RCORE       /REMEMBER PHYSICAL CORE...
          DAP TRPBUF      /...FOR HH TYPEOUT ETC
          LAC C16HH       /HALT HORRIBLE
          JMP DSP9
```

```
/HELD BY XDDT + CRASHED; TYPE ON SOROBAN
XDDTBK,   LAW I STAT
          ADD RSTAT
          RCR 9S
          LAW I 3
          LSM
          JDA 14SOCT
          LAW 77
          AND I (TTNO)
          RCR 6S
          LAW I 2
          JDA 14SOCT
          LAC (DCH XDDTBT)
          JDA 14STXT
          LAC (CORHNG)
          DIP I ERSTAT
HANG1E,   ESM
          JMP HANG1

XDDTBT,   276464
          237756          /XDDT[CR][EOM]

REALHALT, LEM
          LAC (UNUSED)
          DAC I RSTAT
          LAW WHERE-STAT
          ADD RSTAT
          DAP REALH1
          LAW 300
REALH1,   DAC .
          DZM I RCORE
REALH3,   LAW I 1
          LSM
          ADD WTOP
          DAP REALH2
          LIA
          LAW 300
REALH2,   SAS .
          JMP HANG1E
          DIO WTOP
          ESM
          JMP REALH3

/DEBREAK FROM CORE 16 THRU HH16
C16RH,    LAW 77
          AND I (BILLTT)
          JDA TTNUM
          LAW C16RH1
          DAP TTCKSX
          JMP TTCKS1+2
```

```
C16RH1,   SAS (2)
          JMP C16RH2
          LAC C16BKK
          DAC 71
          JMP R0

C16RH2,   LAC (400000)
          IOR I (BITS)
          DAC I (BITS)
          LIO I (HH16AC)
          DIO 70
          LIO I (HH16PC)
          DIO 71
          LIO I (HH16IO)
          DIO 72
          LIO I (HH16FG)
          JMP R0+1

HANG,     RCR 6S          /HANG GUY WITH STATUS IN AC (12-17)
HANG+1,   JSP HANGS       /SUBROUTINE IS IN TT ROUTINES
HANG1,    ISB 1700
          DZM HOTFLG      /TELL SWAPPER
          JMP R0

.RCK,     RCK             /REAL RCK
          JMP DPEEK1

FF8,      SZF I 5
          JMP FF88
          SZF 6
          IOT+743         /LIGHT UP THE SLEEPY OPERATOR
          SZF I 6
          IOT+343
          JMP R1
FF88,     LAC 70          / 8 FLIP-FLOPS
          DCH (ADD DCH .+1)
          IOT+70          /0X=CLEAR ALL,  1X=SET X,  2X=CLEAR X
          JMP R1


DELAY,    IDX 71          /HANG FOR SECONDS (ALSO CALLED FROM CORE 15)
          LAW 70
DELAY2,   DAP DELAY3      /SEE DRPA,DPPA ALSO
          LAC I (BILLTT)
          SCR 9S
          SZA I           /SKIP IF A QUEUE ALREADY SAVED UP
          LAC I ERSTAT
          RCL 9S
          DAC I (BILLTT)
          LAW 7777
DELAY3,   SUB .
          DAP I ERSTAT
          LAW CLKHNG"T"100
          JMP HANG
```

```
/RESTART MODE CONROL IOT
RSMC,     SZF I 5
          JMP RSMC1       /DON'T DEBREAK
          LAC (40000)     /DEBREAK LEGAL BIT
          AND I (BITS)
          SZA I
          JMP ELGIOT
          LAW I 1
          ADD I (RESTPC)
          DAC 71
          LAC (170000
          AND I (RESTPC
          LIA
          SAD (160000
          CLA
          SZA
          JMP ELGIOT
          JSP C16TUS      /UNSAVE CORE 16 IF NECESSARY

RSMC1,    LAC I (BITS)
          AND (-140000)   /RESTART MODE AND DEBREAK LEGAL BITS
          SZF 6
          IOR (100000)    /RESTART MODE BIT
          DAC I (BITS)
          JMP R1

/DISPATCH CLEAR THE READER
DCLR,     JSP RDINIT
          JMP R1          /OWNED BY ANOTHER USER
          JMP R2          /OK

/DISPATCHER RPA
DRPA,     LAW (1)
          SZS 60
          JMP DELAY2
          JSP RDA
          JMP ELGIOT      /UNOWNED BY USER
          JMP . 3
          DIO 72
          JMP R1          /RETURN TO USER WITH CHAR IN IO
          LAW RPAHNG"T"100
          JMP HANG

/DISPATCHER RELEASE THE READER
DRLSRD,   JSP RDRLS
          JMP R1
```

```
/GET READER ROUTINE
XRDIN,    LIF"U"SCF
RDINIT,   DAC RX
          LAC RDWHO
          SPA                 /OR SAD (=0)
          LAC RSTAT
          DAC RDWHO
          SAS RSTAT
          JMP RDIN1
          SZF 1
          JMP RDIN2
          LAC RBO
          DAC RBI
          LAC CRB1
          DAC RB1
RDIN2,    IDX RX
RDIN1,    LFI
          JMP I RX

/GET A CHARACTER FROM THE READER BUFFER
/R1=NOT YOURS, R2=EMPTY, R3=OK, CHAR IN IO
RDA,      DAC RX
          LAC RSTAT
          SAS RDWHO
          JMP I RX
          IDX RX
          LAC RB1
          SAD RB2
          JMP R4              /READER RUNNING
          LAC RBI
          SUB RBO
          RAL 1S
          SPA
          ADD (RDLEN RDLEN)
          SUB (20)
          SMA                 /START READER WHEN 20 CHARS IN BUFFER
          JMP R4
          LAC RB2
          DAC RB1
RB2,      RPA
          RCK
          DIO RBK
R4,       LAC RBO
          SAD RBI
          JMP I RX            /EMPTY
          LIO I RBO
          ADD (400000)
          SAD RBE
          SUB (RDLEN)
          DAC RBO
          SPA
          RIR 9S
          RCR 8S
          CLI
          RCL 8S              / IO BITS 10-17
          IDX RX
          JMP I RX
```

```
RDRLS,      DAC RX          /RELEASE THE READER
            CLI"U"CMI
            LAC RSTAT
            SAD RDWHO
            DIO RDWHO
            JMP I RX

RX,         0

/DISPATCHER GET THE PUNCH
DGETPU,     LAC PUNWHO
            SPA             /OR SAD (=0)
            LAC RSTAT
            DAC PUNWHO
            SAS RSTAT
            JMP R1
            JMP R2

/DISPATCHER RELEASE THE PUNCH
DRLSPU,     CLI"U"CMI
            LAC RSTAT
            SAD PUNWHO
            DIO PUNWHO
            JMP R1

/DISPATCHER PPA
DPPA,       LAC RSTAT
            SAS PUNWHO
            JMP ELGIOT
            LAW (1)
            SZS 60
            JMP DELAY2
            LAC PUP
            IDA
            SAD PEND
            SUB (PUNLEN)
            LIO (PPAHNG
            SAD PSP
            JMP HANG+1
            LIO 72
            DIO I PUP
            DAC PUP
            CLA
            SAS PRUN
            ISB 1200
            JMP R1
```

```
/IOT TO START A USER
PSUC,     ESM
PSU,      LEM
          LAW WHERE-1
          DAP PSUA
PSUD,     IDX PSUA
          SAD (SAS WHERE+100)
          JMP HANG1
          LAW 300
PSUA,     SAS .
          JMP PSUD
          LIO (500)
          LSM
          SAS I PSUA
          JMP PSUC
          DIO I PSUA
          LAW 7777
          AND PSUA
          SAD WTOP
          IDX WTOP
          ESM
          LAW I WHERE-STAT
          ADD PSUA
          DAP PSUB
          LAW 1
PSUB,     DAC .
          DZM HOTFLG
          JMP R1
```

```
/WRITE P POINTER
DWPP,     LAC (-JMP)
          AND 70
          SUB (11)
          SAS (-10)       /OK TO CHANGE JOB NUMBER
          SZS 50
          JMP DWPP1
          SPA
          JMP R1
          SUB (I-11)
          SAD (PRIORITY-I)          /CHANGING HIS PRIORITY?
          JMP DWPP2       /YES
          SMA
          JMP R1
DWPP1,    LAC 72
          DAC I 70
          JMP R1

DWPP2,    LAW I 7777      /SET NEW PRIORITY
          AND 72
          LIA
          DCH (LAC PRIORITY)
          SPI
          JMP R1          /NEG, PRIORITY CANNOT BE THE HIGHEST REACHED
          SUB I (BITS)    /COMPARE WITH PREVIOUS HIGH
          XOR I (BITS)
          RAR 7S
          SWP
          SPI I
          DCH (JMP BITS) /SAVE NEW HIGHEST PRIORITY
          JMP R1

/READ P-POINTER
DPEEK,    LIO I 70
          SZF 1
          LIO I 72
DPEEK1,   SZF 2
          DIO 72
          SZF I 2
          DIO 70
          JMP R1
```

```
/SUPER-PEEK AND SUPER-WPP,  EXAMINE OR CHANGE OTHER USER CORES,
SPP,        LAC SPOWN
            SZA                 /LOGIC ALREADY IN USE?
            JMP SP2             /YES
            LAW SPHANG
            DAP SPX             /SET SWITCH TO HANG OWNER
            LAC 70              /USER AND CAD BEING EXAMINED
            DAP SPCAD
            AND (770000
            RAL 6S
            ADD (STAT
            DAC SPUSER          /STAT PTR TO USER TO BE EXAMINED
            SAD BDSTAT          /UNHANG HIM
            LAW FSTAT
            DAP SP1
            LAW I 7777
            LEM
            LSM
SP1,        AND .
            SAD (UNUSED
            JMP SP3             /USER DOES NOT EXIST
            AND (300000
            CLI"U"SWP
            SNI I
            DIP I SP1
            LAW 777
            AND I SP1           /MASK QUEUE IN CASE CLKHNG
            DAP I SP1
            ESM
            EEM
            LAW 7777
            AND I NETHND
            SAD RSTAT
            JMP SP4             /NETHANDLER DOESN'T NEED SS5
            SZS I 50
            CLF 1
SP4,        DZM SPFLAG          /SET READ-WRITE FLAG,  ZERO IF READING,
            SZF 1
            IDX SPFLAG
            LAW I 1
            DAC SPCNT
            LAC 72              /CONTENTS TO BE WRITTEN
            DAC SPWORD
            SZF 2               /ONE WORD OR MANY WORDS ?
            JMP SP5
SPHANG,     LAC RSTAT
            DAC SPOWN
            LIO (SPPHNG
            SAD SPUSER
            JMP HANG1           /LOOKING AT SELF,  DON'T HANG,
            JMP HANG+1
```

```
SP2,    SAD RSTAT
SPX,    JMP .           /SWITCH,  SPHANG OR SPDONE,

        LAW 7777        /TOO BAD, LOGIC IS BUSY,
        AND I NETHND
        SAD RSTAT       /IS THE "NCP" TRYING TO GET THE LOGIC ?
        JMP SP6
        LAC SPOWN
SP3,    ESM
        DAC 70          /ERROR CODE,  STAT PTR OR "UNUSED" STATUS,
        JMP R1          /ERROR RETURN

SP6,    LAC SPOWN       /TOO BAD FOR NORMAL USER,
        SAD BDSTAT      /THE "NCP" IS USURPING THE LOGIC,
        LAW FSTAT
        IOR (DCH)
        DAC T1
        LAC (300000)
        LSM
        AND I T1
        CLI"U"SWP
        SNI I
        DIP I T1        /UNHANG FORMER OWNER
        LAW 777
        AND I T1
        DAP I T1
        ESM
        DZM SPOWN
        DZM SPUSER
        JMP SPP

SP5,    DZM T1
        DAP T1
        RAL 6S
        CMA
        IOR (777740)
        DAC T3
        DAC SPCNT
        SZF I 1
        JMP SPHANG
        LAW I 40
        ADD SPCAD
        SPA
        JMP ILGIOT
        LAW SPWORD
        DAP .+2
        LAC I T1
        DAC .
        IDX .-1
        IDX T1
        ISP T3
        JMP .-5
        JMP SPHANG
```

```
SPDONE,   DZM SPOWN       /RELEASE SPP LOGIC
          SZF 1
          JMP R2          /GOOD RETURN AFTER WRITING INTO OTHER USER
          SZF 2
          JMP .+4
          LAC SPWORD      /CURRENT CONTENTS
          DAC 70
          JMP R2          /GOOD RETURN, ONE WORD READ

          LAW 7777
          AND 72
          DAC T2
          SUB (40)
          SPA
          JMP ILGIOT
          LAW SPWORD
          DAP .+1
          LAC .
          DAC I T2
          IDX .-2
          IDX T2
          ISP SPCNT
          JMP .-5
          JMP R2          /GOOD RETURN, MANY WORD READ

SPOWN,    0               /0 OR STAT PTR TO OWNER
SPUSER,   0               /STAT PTR TO USER BEING EXAMINED
SPCAD,    0               /CAD WITHIN CORE OF SPUSER
SPWORD,   .+40/           /CONTENTS
SPFLAG,   0               /0 FOR "READ", NON-ZERO FOR "WRITE".
SPCNT,    0               /NUMBER OF WORDS TO TRANSFER

T1,       0               /THREE TEMPORARIES FOR DISPATCHER OR SWAPPER
T2,       0
T3,       0
```

```
/DISPATCHER TT ROUTINES

/SET UP PTRS USED BY DISPATCHER TT ROUTINES
TTSET,    DAP TTSETX
          LAW 37          /ONLY 32. TTYS THESE DAYS
          AND I (TTNO
          JDA TTNUM
TTSETX,   JMP .

TTNUM,    0               /JDA WITH TT NUMBER IN AC
          DAP TTNUMX
          LAW TTP"T"200000
          ADD TTNUM
          SAL 2S          /GET PLACE IN TTP TABLE
          DAP TTFP
          IDA
          DAP TTSP
          IDA
          DAP TTUP
          DAP TTDCHU
          DAP TTLCHU
          IDA
          DAP TTPP
TTNUMX,   JMP .

TTFP,     140000+.        /FLAGS
TTSP,     140000+.        /SERVICE PTR
TTUP,     140000+.        /USER PTR
TTPP,     140000+.        /PROGRAM


/CHECK TT STATUS IOT
TTCKS,    JSP TTCKS1
          SZA I
          JMP TTRETN
          DAC 70          /PUT ERROR CODE IN AC AND ERROR CODE WORD
          JMP TTCKS2      /CLEAR OUT BREAK IF ANY


/SUBROUTINE TO CHECK OWNERSHIP AND BREAK
TTCKS1,   DAP TTCKSX
          JSP TTSET       /TT NUMBER IS IN USER CORE
TTCKS1+2, LAW 377
          AND I TTPP
          LIO (1
          SAS RSTAT
          JMP TTCKSY      /NOT HIS, ERROR CODE 1.
          LAW I 7767      /770010 TO AC
          AND I TTFP
          RAR 4S
          RIL 1S
          SPQ             /BREAK?
          CLI             /NO
TTCKSY,   LAI
TTCKSX,   JMP .
```

```
/SUBROUTINE NORMALLY USED BY TT IOT'S.  CHECKS OWNERSHIP
/AND DETECTS "BREAK" STATUS AND CLEARS IT.  GIVES TRAP (OR R1)
/UNLESS EVERYTHING OK.
TTOK,       DAP TTOKX
            JSP TTCKS1
            SZA I
TTOKX,      JMP .
TTCKS2,     SAS (2
            JMP TTERR       /NOT HIS TT
DHANGX,     LAC I ERSTAT    /BREAK. GET USER'S STATUS
            RAL 3S
            SPA             /"BREAK" BIT SET?
            JMP R0          /BREAK OCCURRED SINCE CH 16 BREAK BEGAN
            LAW 10          /CLEAR OUT BREAK STATUS
            LSM
            AND I TTFP
            SZA I
            DIP I TTFP      /CLEAR NULL COUNT UNLESS BEING INTERRUPTED
            ESM
            LAW 2           /ERROR CODE FOR BREAK
TTERR,      DAC I (ERCODE)  /SAVE ERROR CODE
            LIO 71          /SAVE ADDRESS OF ERROR IN TRAPPC
            DIO I (TRAPPC)
            SZF 6
            JMP R1
            LAW TTTSU
            SPI
            IOR (400000)    /SET OVERFLOW IF PREVIOUSLY ON
            DAC 71
            JMP R0

/IOT TYI.  TYPE IN CHAR TO TOP 6 BITS OF AC.
TYI=.       JSP TTOK
            JSP TTU         /GET CHAR OR HANG
            DAC 70          /STORE CHAR
            JMP TTRETN

/IOT TYO.  TYPE CHAR IN TOP 6 BITS OF AC.
TYO=.       JSP TTOK
            LAW I 7777
            AND 70
            JDA TTS         /TYPE CHAR OR HANG
TTRETN,     SZF 6           /TEST WHETHER 1= OR 2=RETURN IOT

R2,         IDX 71          /NORMAL RETURN 2
R1,         IDX 71          /NORMAL RETURN 1
R0,         LIO DSP1        /NORMAL RETURN 0
R0+1,       LFI             /RESTORE FLAGS
            LAC 70
            LIO 72
            JMP I 71
```

```
/IOT TIS.  TYPE IN STRING TERMINATED BY 74.  AC IS PTR.
TIS,      JSP TTOK
          LAC I (TISMAX
          IDC
          DAC DTEM1
TIS1,     LAC 70          /LOOP.  TEST FOR VALID PTR.
          IDC"U"SCI
          SAD DTEM1
          JMP TIS2        /TISMAX EQUALLED.  WILL BE EXCEEDED.
          AND (177740
          RCL 6S
          SNI"U"SZA I
          JMP TIS3        /TOO LOW OR TOO HIGH
          JSP TTU         /GET CHAR OR HANG
          DCH I 70
          SAD (74
          JMP TTRETN      /TERMINATOR SEEN
          JMP TIS1

TIS2,     LAW 3           /TISMAX ERROR CODE
          JMP TTERR

TIS3,     LAW 4           /ILLEGAL REGION ERROR CODE
          JMP TTERR

/IOT TOS.  TYPE STRING TERMINATED BY 74.  AC IS PTR.
TOS,      JSP TTOK
TOS1,     LIO 70
          LCH I 70        /GET CHAR
          SAD (740000
          JMP TTRETN      /RETURN, LEAVING AC STEPPED.
          DIO 70          /UNSTEP IN CASE OF HANG
          JDA TTS         /TYPE CHAR OR HANG
          LCH I 70        /STEP AC
          JMP TOS1

/SUBROUTINE TO TAKE 6-BIT CHAR OUT OF BUFFER.  HANG USER IF EMPTY.
TTU,      DAP TTSX        /EXIT THROUGH TTS
          ERG
          LAW I 2
          DSC 600
          AND I TTFP
          SAS I TTFP
          JMP TTUFUL      /PF 5 SET.  BUFFER FULL WITH PTRS EQUAL.
          RCR 6S          /SAVE PF 1 IN I.O.
          LAC I TTUP
          SPI I           /TYPEACTIVE?
          SAD I TTSP      /EMPTY?
          JMP TIHANG      /TYPEACTIVE OR BUFFER EMPTY
TTU1,     IDC             /STEP USER PTR
          DAC I TTUP
          JMP TTS5

TTUFUL,   DAC I TTFP      /CLEVERLY CLEARS PF 5
          LAC I TTUP
          JMP TTU1
```

```
/SUBROUTINE TO TYPE 6 BITS IN TOP OF AC.  HANG USER IF BUFFER FULL.
TTS,        0
            DAP TTSX
            LAW I 7777
            SAD TTS
            JMP TTS7        /WARNING CHAR.  GOBBLE IT UP.
            ERG
            LAW 40
            LIO I TTUP
            DSC 600
            IOR I TTFP      /STF 1
            SAS I TTFP      /ALREADY SET?
            JMP TTSBEG      /NOT TYPEACTIVE.  START HIM UP.
TTS1,       LAI
            IDC
            SAD I TTSP
            JMP TOHANG      /BUFFER FULL
            LIO I TTPP
            RIL 8S
            SPI I
            JMP TTS2        /77 NOT SAVED UP
            IDC
            SAD I TTSP
            JMP TOHANG      /NO ROOM FOR 2-BYTE CHAR
            CLC
            XCT TTDCHU      /STORE 77
TTS2,       LAC TTS
TTDCHU,     DCH I .         /STORE CHAR
TTS3,       LAW I 1000
            AND I TTPP
TTS4,       DAC I TTPP      /SET OR CLEAR "SAVED 77" STATUS
TTS5,       JSP ASC6        /RE-ACTIVATE CH 6
TTLCHU,     LCH .           /GET CHAR FOR TTU RETURN
            LRG
TTSX,       JMP .

TTSBEG,     DIO I TTSP      /SET PTRS EQUAL
            AND (-2         /CLF 5
            DAC I TTFP
            AND (211        /KEEP LINK, PF 3, AND PF 6
            SAS (1
            JMP TTS1        /EXPECTING A BREAK; DON'T TCB
            LIO TTNUM
            SSB
            LAW TTTBL"T"10000           /CALCULATE PLACE IN TBL
            IOR TTS
            RAL 6S
            LIO I TTPP
            RIL 8S
            SPI             /"SAVED 77" KEPT IN "1000" BIT
            ADD (100        /77 SAVED UP
```

```
            DAP . 1
            LIO .           /GET CODE
            TCB             /SEND IT
            SAS (TTTBL+115
            SAD (TTTBL+100
            JMP TTS6        /BREAK OR CR-NO-LF,  SET LINK,
            JMP TTS3

TTS6,       LAW 200         /SET LINK
            IOR I TTFP
            DAC I TTFP
            JMP TTS3

TTS7,       LAW 1000        /SET "SAVED 77" STATUS
            IOR I TTPP
            JMP TTS4

/ROUTINES TO HANG USER, ETC,
TIHANG,     LIO (TYIHNG
            JMP TTHANG

TOHANG,     LIO (TYOHNG
TTHANG,     LRG
            IDX TTSX        /IN CASE XDDT
            SZL
            JMP TTSX        /XDDT,  GIVE HIM R2,
GTYBSX,     SZF 5
            JMP TTHNG1      /A "DON'T HANG ME" IOT
            JSP HANGS       /HANG USER
            JSP ASC6        /REACTIVATE CHANNEL
            JMP HANG1

/SUBROUTINE TO HANG USER,  HUNG STATUS IN I,O,
HANGS,      DAP HANGSX
            LAW I 7777      /LOOK AT TOP 6 BITS OF...
            LSM
            AND I ERSTAT    /...GUY'S STATUS,
            SWP
            SNI
            DIP I ERSTAT    /NO SPECIAL BITS SET; HANG HIM,
            ESM
HANGSX,     JMP .

/SUBROUTINE TO RE-ACTIVATE CH 6 AND ISB IF SCANNER STOPPED,
ASC6,       DAP ASC6X
            ASC 600
            CKS
            RIR 2S
            SPI
            ISB 600         /SCANNER STOPPED
ASC6X,      JMP .
```

```
/IOT'S TO LEAVE, ENTER, AND SAVE UP 8-BIT MODE AND CONTROL MODE.
TTMODE,   JSP TTCKS1
          SAD (1
          JMP TTERR       /NOT HIS
          LAC I TTFP
          SZF 3
          DAC 72          /RETURN OLD STATUS IN I.O.
          LAW 4
          SZF I 1         /8-BIT IOT?
          JMP TTMOD1      /NO
          LAC 70          /YES.  SET NEW ALARM CHAR.
          XOR I TTPP
          AND (776000     /ALARM CHAR IN TOP 8 BITS
          XOR I TTPP
          LIO I TTPP
          DAC I TTPP
          SZF 3
          DIO 70          /PUT OLD ALARM CHAR IN AC
          LAW 20
TTMOD1,   DAC DTEM1       /STORE FLAG BIT FOR MODE
          CMA
          LSM
          AND I TTFP
          SZF 2           /ENTER MODE?
          IOR DTEM1       /YES
          DAC I TTFP
TTMODX,   ESM
          JMP TTRETN


/IOT "GET TELETYPE"
GTY,      JSP TTSET
          LIO RSTAT
          LAW 377
          LSM
          AND I TTPP
          SZA I
          DIO I TTPP      /IF NO ONE'S: ASSIGN TO HIM
          ESM
          SZA
          SAD RSTAT
          JMP TTRETN      /NORMAL RETURN
TTHNG1,   CLC             /"DON'T HANG ME" CODING
          DAC 72          /=0 IN I.O. TO INDICATE HANG CONDITION
          JSP ASC6        /REACTIVATE CHANNEL 6
          JMP TTRETN      /GIVE HIM A GOOD RETURN
```

```
/IOT RELEASE TT
RTY,        JSP TTCKS1
            SAD (1)         /NOT YOUR TT?
            JMP TTERR
            DZM HOTFLG
            LAW 7741
            LSM
            AND I TTFP
            DAC I TTFP
            DZM I TTPP
            ESM
            JMP TTMODX

RTY1,       DCH .

/IOT TO HANG USER UNTIL BREAK ON TT IN TTNO
DHANG,      DSC 600
            JSP TTCKS1
            LIO (EOTHNG)
            SAS (2)         /BREAK?
            JMP TTHANG      /NO, HANG HIM
            JSP ASC6
            JMP DHANGX      /IN TTOK, CLEAR OUT BREAK AND UNHANG
```

```
/EXEC DDT TYO AND TYI
XDDTYO,   JDA XDDTX       /CALLED FROM CORE 10
          CLA
          RCR 6S
          JDA TTS         /TYPE FROM BOTTOM OF IO
          JMP XDDTXR
XDDTXY,   LAI
          LIO DDTFLG
          LFI
          LIA
          JSP HNGDDT
XDDTX1,   CLL"U"CML       /XDDT FLAG FOR TT ROUTINES
          LAC EXECTT
          JDA TTNUM       /SETUP POINTERS
          LIO XDDTEM
XDDTXX,   JMP .

XDDTX,    0               /EXIT ACROSS CORE
          DAP XDDTXX
          DIO XDDTEM
          LIF"U"SCF
          DIO DDTFLG
          JMP XDDTX1

XDDTXR,   LIO DDTFLG      /RETURN ACROSS CORE
          LFI
          LIO XDDTEM
          JMP I XDDTX

XDDTYI,   JDA XDDTX       /CALLED FROM CORE 10
          JSP TTU         /TYPE INTO TOP OF AC
          JMP XDDTXR
          DZM DDTGUY
          JMP XDDTXY

XDDTEM,   0
EXECTT,   0
DTEM1,    0
```

```
/ROUTINE FOR DISPATCHER TO HANG XDDT
HNGDDT,   DAC HDDTX
          LAW I 7777
          LSM
          AND STAT 0      /XDDT IS USER 0
          SWP
          SNI
          DIP STAT 0
          ESM
          JSP ASC6        /RE-ACTIVATE CH 6 AFTER HANGING DDT
HNGDD2,   LIF
          DIO DDTFLG
          JMP SWORG       /ON CH 17 BREAK = CALLED FROM XDDT

/ROUTINE TO START UP XDDT WHEN IT COMES INTO CORE
RUNDDT,   LAW I 7777
          LSM
          AND STAT 0
          SZA
          JMP DDTBRK      /BREAK KEY ON EXEC TT
RUNDD1,   ESM
          LAC RCORE
          DAC DCORE
          SAR 2S
          IOR (200000)    /XDDT RUNS IN CORE 10 (RENAMED)
          DCH (ADD .+1)
          RNM
          LAC DDTGUY
          SZA             /DID HE HAVE ANYBODY
          JMP DDTWT1      /YES, GET HIM INTO CORE AGAIN
RUNDD2,   EEM
          LIO DDTFLG
          LFI
          JMP I HDDTX     /NO, GO BACK TO DDT'S CORE

HDDTX,    100000
DDTGUY,   0
DDTSU,    100000
DDTNUS,   100001
DDTFLG,   0

DDTBRK,   LAW 1
          DAC STAT 0
          LAC DDTSU
          DAC HDDTX
          DZM DDTGUY
          JMP RUNDD1
```

```
/ROUTINE FOR XDDT TO CALL TO GET A GUY INTO CORE
DDTWNT,   DAC HDDTX
          LEM
          LAW STAT
          AAI
          DAC DDTGUY
DDTWT1,   SAD BDSTAT
          LAW FSTAT
          DAP DDTWT3
          LAW WHERE-STAT
          ADD DDTGUY
          DAP DDTWT2
          LAW 777
          LSM
DDTWT2,   AND .
          SAD (300        /IS THERE SUCH A GUY
          JMP DDTHLT      /NO
          LAC (300000
DDTWT3,   AND .
          CLI"U"SWP
          SNI
          LAC I DDTWT3
          IOR (10000)
          DIP I DDTWT3    /SET BIT
          LAW 777         /MASK QUEUE IN CASE GUY WAS CLKHNG
          AND I DDTWT3
          DAP I DDTWT3
          ESM
          LAW I 2
          DAC I DCORE     /HOLD DDT'S CORE
          JMP HNGDD2

/ROUTINE TO START UP XDDT WHEN HIS USER COMES INTO CORE
DDTWS,    LAC (-10000)    /REMOVE XDDT WANTS BIT
          LSM
          AND I RSTAT
          DIP I RSTAT
          ESM
          LAC RSTAT       /IS THIS GUY DDT CURRENTLY WANTS
          SAS DDTGUY
          JMP SWORG       /NOT GUY; FORGET IT
          SAD SPOWN
          DZM SPOWN       /IN CASE DDT CHANGES HIS PC
          LAW STAT 0      /IS GUY; SET DDT RUNNING
          DAC I DCORE
          DAC RSTAT
          DAP ERSTAT
          JMP RUNDD2

DDTHLT,   ESM
          LAC DDTNUS      /TELL DDT THAT THERE'S NO USER
          DAC HDDTX
          JMP RUNDD2
```

```
/XDDT RPA
DDTRPA,    DAC DDTRPX
DDTRP1,    JSP RDA
           XX
           JMP DDTRDH
           JMP I DDTRPX
DDTRPX,    0

DDTRDH,    LIO (RPAHNG)
           JSP HNGDDT
           JMP DDTRP1

START
```