MAG TAPE SYSTEM (MAGTAP,24) 2/1/72 BPC

```
TGO,        Ø                   /ENTRY IF BITS TO BE OR'ED IN
TGO+1,      DAP 4CHAN3
            LAW I 4
            DAC TERCNT
TGOR,       LAC I 4CHAN3
            IOR UNIT
            IOR TGO
            LIA
            IDX 4CHAN3
            RRO 2ØØ
            JSP THLT1
            JMP I CH4BRK

TGO1,       DZM TGO
            JMP TGO+1
THLT1,      LAW I 1
            ADD 4CHAN3
            DAC 4CHAN3
            ISP TERCNT
            JMP TGOR
            JMP THLT

TERCNT,     Ø
↑L
```

```
/CHANNEL 4 TAPE SERVICE ROUTINE

4CHAN1:   LIO CONTR2
          SPI
          JMP I CH4BRK
          RIL 1S
          SPI I
          JMP I CH4BRK
          RRI 12ØØ
          LAC I (DCH 72) /WASTE 15 MICRO-SECONDS
          RIR 8S
          SPI
          JMP I CH4BRK
          RRI 2ØØ
          SPI I
          IDX 4CHAN3
          JMP I 4CHAN3

4CHAN3.   DCH I FTAPE    /ALWAYS IN CORE 15

/TAPE WAIT ROUTINE
7WAIT.    DAC 7WAITX
          LIO UNIT
          LAW ØUNIT
          SNI I
          LAW 1UNIT
          DAP 7WAIT1
          LAC I CONTR2
          XOR I 7WAITX
          AND (7777
          XOR I 7WAITX
7WAIT1.   DAC .
          IDX 7WAIT1
          IDX CONTR2
          LAC I CONTR2
          XCT 7WAIT1
          IDX 7WAIT1
          IDX CONTR2
          LAC I CONTR2
          XCT 7WAIT1
          LAC ØCTR
          AAI             /UNIT STILL IN IO
          DAC 7WAITX
          LAW I 1
          DAC I 7WAITX
          IDX CONTR2
          JMP FTAPE5

7WAITX.   Ø
1SECX.    Ø
ØUNIT.    REPEAT 3.   Ø
1UNIT.    REPEAT 3.   Ø
↑L
```

```
/CHANNEL 7 TAPE SERVICE ROUTINE
1SECJA.   DAC 1SECX
          LIO END2
          LAC ØUNIT
          DIO ØUNIT       /USE ØUNIT AS TEMP STORAGE
          DAC I ØUNIT
          IDX ØUNIT
          LAC ØUNIT+1
          DAC I ØUNIT
          IDX ØUNIT
          LAC ØUNIT+2
          DAC I ØUNIT
          IDX ØUNIT
          JMP 1SECJC

1SECJB.   DAC 1SECX
          LIO END2
          LAC 1UNIT
          DIO 1UNIT
          DAC I 1UNIT
          IDX 1UNIT
          LAC 1UNIT+1
          DAC I 1UNIT
          IDX 1UNIT
          LAC 1UNIT+2
          DAC I 1UNIT
          IDX 1UNIT
1SECJC.   SAD (TTBL+6
          LAC (TTBL
          LSM
          DAC END2
          LAC CONTR2
          IOR (200000
          DAC CONTR2
          ESM
          ISB 400
          JMP I 1SECX
↑L
```

```
/TAPE DISPATCH ROUTINE
BFTAPE:
FTAPE.      NOP
            LAC CONTR2
FTAPE+2.    AND (177777)
            SAD END2
            JMP FTAPE6
            DAC FTAPE1
            CLA
            LIO I FTAPE1
            RCL 6S
            ADD (FTAPET
            DAP FTAPE2
            CLA
            RCL 1S
            DAC UNIT
            SAL 1S
            ADD (UNTBL
            DAC UNPTR
            CLA
            RCL 2S
            RAR 8S
            DAC TAPSTT
            RIR 9S
            DIO I TUSER
            IDX FTAPE1
            LAC I FTAPE1
            DAC TAPPAR
            IDX FTAPE1
            DZM ERWRD2
            CLI
            LAC I FTAPE1
            RCL 6S
            LFI
            LAC TAPPAR
FTAPE2.     XCT .
↑L
```

```
/USER WANTS DATA CHANNEL

TPRLS1.   LAC I UNPTR     /MARK AS NO LONGER AT LOAD POINT
          RAL 2S          /ARE WE IN FORWARD POSITION
          SMA
          JMP TPRLS2      /IN FORWARD POSITION
          JSP TGO1
          204000
          JSP THLT
TPRLS4.   JSP 7WAIT
          040000

TPRLS3.   JSP TGO1
          200000
          JMP TPRLS4
          RRI 1200
          LAC I (DCH 72) /WASTE 15 MICROSECONDS
          RIL 2S
          SPI
          JMP TPRLS4
          LAC I UNPTR     /MARK IN FORWARD POSITION
          XOR (100000
          DAC I UNPTR

TPRLS2.   LAC (550000     /SET TO USE CHANNEL
          LSM
          DIP CONTR2
          CLA             /IS CHANNEL BUSY
          SAD I DCONT
          ISB 100
          ESM
FTAPE3.   LAW FTAPE
          DAP 4CHAN3
          JMP I CH4BRK

FTAPE6.   DAC CONTR2
          JMP FTAPE3

FTAPET.   NOP
          JMP 4SPC
          JMP 4RWD
          JMP 4ENT
          JMP TPRLS3
          JMP LDPT
          JMP LDPT1
          JMP 4WEOF
↑L
```

```
/RELEASE USER ON CHANNEL 4

TPRLS.    LAC I TUSER
          SUB MSTAT
          DAP FTAPE4
          LAC (377777
          AND I FTAPE4
          DAC I FTAPE4
          ISB 1700
          DZM I HOTFLG
          LAW 3
          ADD CONTR2
FTAPE5.   AND (177777
          SAD (TTBL+6
          LAC (TTBL
          IOR (200000
          DAC CONTR2
          JMP FTAPE+2

FTAPE4.   DCH .
FTAPE1.   0
UNIT.     0
TAPPAR.   0
UNPTR.    0
TAPSTT.   0

/CHANNEL 16 TAPE INITIALIZE ROUTINE
/LINK SET←RETURN 3

16TAPE.   DAP 16TAPX
          JSP I .SETUP
          LAC I (DCH 72)
          LIA
          AND (-1)
          SZA
          JMP NOTYOURS
          JSP 16TAP2
          AND (200177
          SAS I 16USER
          JMP NOTYOURS
          LIO I 16TAP1
          LAC I (DCH 70)
16TAPX.   JMP .

16TAP1.   0
16UNIT.   0
↑L
```

```
/TAPE WAITING LIST ROUTINE

TAPE,     Ø
          DAC TAPE3       /SAVE POINTER TO COMMAND
          JSP I RELEASE
          LAC I TAPE3
          IOR 16UNIT
          IOR I 16USER
          DSC 7ØØ         /DE-ACTIVATE CHANNEL 7 SO END2 WILL NOT CHANGE
          LIO END2        /GET END OF RING BUFFER
          DIO TAPE1
          DAC I TAPE1     /SAVE FIRST WORD
          IDX TAPE1
          LAC TAPE
          DAC I TAPE1
          IDX TAPE1
          LIF
          RCR 6S
          DIP TAPE3
          IDX TAPE3       /GET RETURN ADDRESSS
          DAC I TAPE1
          IDX TAPE1
          SAD (TTBL+6)
          LAC (TTBL)
          LSM
          DAC END2
          LAC CONTR2
          IOR (2ØØØØØ)
          DAC CONTR2
          ASC 7ØØ
          ESM
          ISB 4ØØ
          JMP I RØ

TAPE1,    Ø
TAPE3,    Ø
CONTR2,   TTBL
END2,     TTBL

16TAP2,   DAP 16TP2X
          DIO GTAPE7
          RIR 7S
          LAI
          SZF 4
          IOR (2ØØØ)
          SZF 5
          IOR (1ØØØ)
          DAC 16UNIT
          RIL 8S
          LAI
          ADD (UNTBL)
          DAC 16TAP1
          LAC I 16TAP1
16TP2X,   JMP .
↑L
```

```
/REWIND ICT
/FLAG 1=1←WITH INTERLOCK. Ø←WITHOUT INTERLOCKTL

RWD:    JSP 16TAPE
        JDA TAPE
        Ø2ØØØØ

4RWD.   SZF 1
        JMP RWD2        /WITH INTERLOCK
        JSP TGO1
        2ØØØØØ
        JSP THLT1
        RRI 12Ø0
        LAC I (DCH 72) /WASTE 15 MICROSECONDS
        RIL 8S
        SPI
        JMP LDPT4
        JSP TGO1
        21ØØØØ
        JSP THLT
LDPT3.  JSP 7WAIT
        Ø5ØØØØ          /WAIT FOR LOAD POINT

RWD2.   JSP TGO1
        212ØØØ
        JSP THLT1
        LAC (2ØØØØØ
        DAC I UNPTR
        IDX UNPTR
        CLC
        DAC I UNPTR
        JMP TPRLS
↑L
```

```
/SPACE TAPE
/FLAG 1=1←BACKWARDS, Ø←FORWARDS
/FLAG 2=1←FILES, Ø←BLOCKS (FILES BACKWARDS IS ILLEGAL)

SPC:    JSP 16TAPE
SPC1,   SPA
        JMP 16ILLSPC
        SZA I
        JMP I RØ
        JDA TAPE
        Ø1ØØØØ

4SPC,   SZF I 2
        JMP SNB         /SPACE BLOCKS
        DAC ERWRD2
        CMA
        DAC TAPPAR
SNF4,   CLF 3
SNF1,   LAW 7777
        JDA .SFNB
        JMP SNF1
        SZF I 3         /EOF-WAS IT FIRST
        JMP SNF3
        LAC .SFNB
        SAD (7776
        JMP SNF5
        ISP TAPPAR
        JMP SNF1
        JMP TPRLS

SNF5,   LAC ERWRD2
        ADD TAPPAR
        JMP SNF6

/FIRST END OF FILE
SNF3,   STF 3
        ISP TAPPAR
        JMP SNF1
        JMP TPRLS

SNB,    JDA .SFNB
        JMP TPRLS       /GOOD RETURN
        LAW 1
        ADD .SFNB
SNF6,   DAC ERWRD2
4EOF,   LAW 2ØØ
        JDA TAPERR
↑L
```

```
/SUBROUTINE TO SPACE FORWARD N BLOCKS
.SFNB.      0
            DAP .SFNBX
            LAC (260000
            SZF I 1
            JMP .SFNB5
            LAC I UNPTR
            IOR (100000)
            DAC I UNPTR
            LAC (270000)
.SFNB5.     DIP .SFNB3+2    /SET WHETHER FORWARD OR BACKWARDS
.SFNB1.     LAC .SFNB
.SFNB1+1.   SPQ
.SFNBX.     JMP .           /GOOD RETURN
            SUB (77)
            DAC .SFNB
            SPA
            JMP .SFNB2      /LAST TIME
            LAW 1760
.SFNB3.     IOR TAPSTT
            JDA TGO
.SFNB3+2.   0
            JMP .SFNB4
            JMP .SFNB1

.SFNB4.     RRI 200
            LAI
            RRI 1200
            AND (77)        /RESET BLOCK COUNT
            ADD .SFNB
            DAC .SFNB
            RIL 8S          /HIT LOADPOINT?
            SPI
            JMP 4LDPT
            RIL 1S          /PERHAPS AN EOT
            SPI
            JMP 4EOT
            RRI 200         /CHECK FOR DATA ERROR
            LAI
            RAL 2S
            SPA
            JMP THLT        /UNRECOVERABLE
            AND (70000)
            SZA
            JMP .SFNB1      /RE-EXECUTE FOR DATA ERRORS
            IDX .SFNBX
            JMP .SFNBX

4EOT.       LAW 1000
            JDA TAPERR

.SFNB2.     ADD (77)
            SAL 4S
            DZM .SFNB
            JMP .SFNB3
↑L
```

```
/WRITE END OF FILE
WEOF:   JSP 16TAPE
        RIL 3S          /TAPE STATUS IN IO
        SPI I
        JMP WRING
        JDA TAPE
        070000

4WEOF,  CLF 2
        LAW I 4
        DAC TAPPAR      /ERROR COUNT
WEF2,   LAC TAPSTT
        AND (-2000)
        JDA TGO
        230000
        JMP WEFERR
        JMP TPRLS

WEFERR, RRI 200
        LAI
        AND (16000
        SZA
        JMP WEFER1      /DATA ERROR
        RRI 1200
        LAC I (DCH 72) /WASTE 15 MICROSECONDS
        RIL 9S
        SPI
        JMP 4EOT
        RIL 2S
        SPI I
        JSP THLT
WEFER1, LAC TAPSTT
        JDA TGO
        270000
        JSP THLT1
        ISP TAPPAR
        JMP WEF2
        SZF 2
        JSP THLT        /DATA ERROR
        JSP TGO1
        216000
        JSP THLT
        STF 2
        LAW I 4
        DAC TAPPAR
        JMP WEF2
↑L
```

```
/ERASE TAPE
ERASE:    JSP 16TAPE
          RIL 3S
          SPI I
          JMP WRING
          SZA I
          JMP I RØ
          SPA
          JMP 16ILLSPC
          JDA TAPE
          Ø3ØØØØ
4ENT.     CMA
          DAC TAPPAR
4ENT1,    JSP TGO1
          216ØØØ
          JSP THLT1       /NO ERRORS WE CAN DO ANYTHING ABOUT
          RRI 12ØØ        /EXCEPT EOT (WHICH ISN'T AN ERROR)
          LAC I (DCH 72)  /KILL SOME TIME (15 MICROSECONDS)
          RIL 9S
          SPI
          JMP 4EOT
          ISP TAPPAR      /COUNT 5 INCHES
          JMP 4ENT1       /AND GO BACK FOR MORE
          JMP TPRLS
↑L
```

```
TPRT.     DAP TPRTX       /TYPE OUT FOR MAGNETIC TAPE
          LIO I 16USER    /TYPE OUT PROG. NO.
          RIL 9S
          LAW I 3
          LSM
          JDA SOTOCT
          LIO I (BILLTT) /TYPE OUT TT.
          RIR 6S
          LAW I 2
          JDA SOTOCT
          LAC (TTYP1)
          JDA SOTXT       /TYPE FIRST PART OF MESSAGE
          LAC I (DCH 72)
          SPA
          CMA
          JDA SOTPRT
          LAC (TTYP2)
          JDA SOTXT       /TYPE SECOND PART OF MESSAGE
          LIO GTAPE7
          RIR 3S
          LAW I 1
          JDA SOTOCT      /TYPE UNIT NO.
          LIO I (DCH 72)
          LAC (TTYP3)
          SPI
          JDA SOTXT
          CLC             /TYPE CARRIAGE RETURN
          JDA SOTDCH
          ESM
TPRTX.    JMP .
↑L
```

```
/TAPE CONTROL IOT'S
TPCNT:  SZF 1           /GET OR RELEASE TAPE
        JMP RTAPE
GTAPE.  JSP I .SETUP
        CLI
        LAC I (DCH 72) /SEE IF REEL WANTED IS ALREADY ON TAPE DRIVE
        SPA
        CMA
        SAD .UNTBL+1    /UNIT Ø
        JMP GTAPE1      /YES-GO SEE IF AVAILABLE
        LIO (1
        SAD .UNTBL+3    /UNIT 1
        JMP GTAPE1      /YES-GO SEE IF AVAILABLE
        CLA             /NOT ON TAPE DRIVE-IS THERE A USEABLE DRIVE
        DZM GTAPE7
        LIO .UNTBL      /CHECK STATUS OF DRIVE Ø
        RCL 2S
        SAD (1          /IS IT USEABLE AND FREE
        JMP GTAPE4      /YES
        IDX GTAPE7      /LOOK AT DRIVE 1
        LIO .UNTBL+2
        CLA
        RCL 2S
        SAS (1          /IS UNIT 1 USEABLE AND FREE
        JMP EMPTY       /NO DRIVES AVAILABLE

GTAPE4. LIO GTAPE7
        JSP 16TAP2
        JSP TPRT        /TYPE OUT ON SOROBAN
        JSP GSET
        IDX 16TAP1
        LAC GTAPE7
        LIO I (DCH 72)
        DAC I (DCH 72) /PUT UNIT IN IO
        LAC I 16TAP1
        SPI
        CMI
        DIO I 16TAP1    /SAVE REEL NUMBER
        SAS (-3)
        JMP GTAPE9
        LAC I (DCH 70)
        JDA TAPE
        050000

GTAP14. SIR 1S
        DIO I (DCH 72)
        SZF I 3
        JMP I R4
        LAC I (DCH 70)
        JMP SPC1

GTAPE7. Ø
↑L
```

```
LDPT1.    JSP TGO1          /REWIND WITH INTERLOCK
          212000
          JSP THLT1         /ERROR IS UNFORGIVEABLE
LDPT.     JSP TGO1          /NOP TO CONNECT TO CHANNEL
B1.       200000
          JMP LDPT3         /NOT ACCEPTING COMMANDS YET
LDPT2.    LAC .             /MARK AS AT LOAD POINT AND WRITE RING IN
          AND I UNPTR       /ALSO MARK FORWARD POSITION
          IOR .
          DIP I UNPTR
          RRI 1300          /TAPE FAULT DIFFERENT IF WRITE RING OUT
          LAC I (DCH 72)    /WASTE 15 MICROSECONDS
          LAI
          RRI 1200
          AND (-0
          XAI
          RAL 1S
          SMA
          JMP LDPT4
          LAC (400000)      /REMOVE BIT
          XOR I UNPTR
          DIP I UNPTR
LDPT4.    SZF I 3
          JMP TPRLS
          LAC TAPPAR        /IS IT LEGAL COUNT
          SPA
          JMP 4ILLSPC
          SZA I             /SPACE 0?
          JMP TPRLS         /YES. DONE
          JMP 4SPC

/TAPE ON DRIVE-SEE IF AVAILABLE
GTAPE1.   JSP 16TAP2
          XOR I 16USER      /CHECK IF ALREADY OWNED
          AND (7777)
          SZA I
          JMP GTAP15
          LAC I 16TAP1      /NOT ALREADY YOURS IS IT FREE
          RAL 1S
          SMA
          JMP NOTYOURS      /DRIVE ALREADY TAKEN-CAN NOT HAVE
          JSP GSET
GTAP15.   LAC I 16TAP1      /YOURS: DO YOU WANT TO CHANGE
          RAL 3S
          XOR I (DCH 72)
          SMA
          JMP GTAP14        /NO: RETURN TO USER
          JSP TPRT          /TYPE OUT ON SOROBAN
          LIO GTAPE7
          DIO I (DCH 72)
GTAPE9.   LAC I (DCH 70)
          JDA TAPE
          060000            /WAIT FOR LOAD POINT
↑L
```

```
/RELEASE TAPE
RTAPE.    JSP 16TAPE      /GO CHECK IF YOUR UNIT ETC.
          LAW I 7777
          NAI
          IOR (200000)    /MARK AS NOT HELD
          DAC I 16TAP1
          CLF 1
          JSP TAPE+1
          20000           /REWIND TO LOAD POINT

GSET.     DAP GSETX
          LAC I 16USER
          XOR I 16TAP1
          AND (207777)
          XOR I 16TAP1
          DAC I 16TAP1
GSETX.    JMP .
TTYP1:    444523          005600
TTYP2:    004645          005600
TTYP3:    005156
↑L
```

```
/READ PHYSICAL RECORD
/AC=CORE LOCATION
/IO=UNIT NO.
/WORD AFTER IOT IS LENGTH OF RECORD

RPR:    SZF 3
        STF 1
        SZF I 1
        CLL↑UCML        /MARK AS RETURN THREE IOT
        JSP 16TAPE
        SZF I 2         /IS IT A WRITE
        JMP RPR6        /NO
        RIL 3S          /YES-IS WRITE RING IN
        SPI I
        JMP WRING
RPR6,   LAW 1
        ADD I RPC       /GET POINTER TO RECORD LENGTH
        DAC TAPE        /UUSE AS TEMP STORAGE
        LAW 28.         /CHECK IF BCD : LENGTH 28, IF IMPLIED
        SZF 4
        LAW 200.+10
        SZF I 1
        LAC I TAPE      /LENGTH NOT IMPLIED
        AND (7777)      /SET TO TWELVE BITS
        JDA TAPE
        000000
        CLF 1           /RESET FLG 1 FOR IOPMAX CHECK
        LAC TAPPAR      /IS IT A ZERO LENGTH BLOCK
        SZA I
        JMP I SWAP      /DO NOTHING
        SZF 2
        JMP RPR5        /IF WRITE SKIP IOPMAX CHECK
        LAC I AC
        SUB BOUND       /CHECK FOR PROTECTION ERROR
        SPA
        JMP ILLSPC
        ADD BOUND
        ADD TAPPAR      /ADD IN LENGTH
        SUB (I)
        SZM
        CLA↑USTF 1
        ADD (I)
        SUB I AC
↑L
```

```
RPR5,       DAP WPRC4
RPR5+1,     LAW 7777
            AND I AC
            IOR I SCORE
            DAC WPRC3       /SET DAC COMMAND
RPR2,       LAC UNIT
            IOR TAPSTT
            IOR (240000
            SZF 2
            XOR (60000)
            DAC WPRC2
            JSP I GO
            I WPRC1
            RRI 700
            LAC (JMP)
            IAI
            SUB WPRC3
            DAC I ERCOD2
            SZF I 1
            JMP I SWAP
MAXERR,     LAC (100000)
            JMP 1TERR

WPRC1,      560002
WPRC2,      220000
WPRC3,      600000
WPRC4,      140000
            000000

START
↑L
```